# NAG Library Routine Document

# F07BUF (ZGBCON)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07BUF (ZGBCON) estimates the condition number of a complex band matrix $A$, where $A$ has been factorized by F07BRF (ZGBTRF).

## 2 Specification

```
SUBROUTINE F07BUF (NORM, N, KL, KU, AB, LDAB, IPIV, ANORM, RCOND, WORK,         &
                   RWORK, INFO)

INTEGER                 N, KL, KU, LDAB, IPIV(*), INFO
REAL (KIND=nag_wp)      ANORM, RCOND, RWORK(N)
COMPLEX (KIND=nag_wp) AB(LDAB,*), WORK(2*N)
CHARACTER(1)            NORM
```

The routine may be called by its LAPACK name ***zgbcon***.

## 3 Description

F07BUF (ZGBCON) estimates the condition number of a complex band matrix $A$, in either the 1-norm or the $\infty$-norm:

$$\kappa_1(A) = \|A\|_1 \left\|A^{-1}\right\|_1 \qquad \text{or} \qquad \kappa_\infty(A) = \|A\|_\infty \left\|A^{-1}\right\|_\infty.$$

Note that $\kappa_\infty(A) = \kappa_1\left(A^{\mathrm{H}}\right)$.

Because the condition number is infinite if $A$ is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine should be preceded by a call to F06UBF to compute $\|A\|_1$ or $\|A\|_\infty$, and a call to F07BRF (ZGBTRF) to compute the $LU$ factorization of $A$. The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\left\|A^{-1}\right\|_1$ or $\left\|A^{-1}\right\|_\infty$.

## 4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

## 5 Parameters

1: NORM – CHARACTER(1) *Input*

*On entry*: indicates whether $\kappa_1(A)$ or $\kappa_\infty(A)$ is estimated.

NORM = '1' or 'O'
$\kappa_1(A)$ is estimated.

NORM = 'I'
$\kappa_\infty(A)$ is estimated.

*Constraint*: NORM = '1', 'O' or 'I'.

2: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: N $\geq 0$.

3: KL – INTEGER *Input*

*On entry*: $k_l$, the number of subdiagonals within the band of the matrix $A$.

*Constraint*: KL $\geq 0$.

4: KU – INTEGER *Input*

*On entry*: $k_u$, the number of superdiagonals within the band of the matrix $A$.

*Constraint*: KU $\geq 0$.

5: AB(LDAB,$*$) – COMPLEX (KIND=nag_wp) array *Input*

**Note**: the second dimension of the array AB must be at least $\max(1, \text{N})$.

*On entry*: the $LU$ factorization of $A$, as returned by F07BRF (ZGBTRF).

6: LDAB – INTEGER *Input*

*On entry*: the first dimension of the array AB as declared in the (sub)program from which F07BUF (ZGBCON) is called.

*Constraint*: LDAB $\geq 2 \times \text{KL} + \text{KU} + 1$.

7: IPIV($*$) – INTEGER array *Input*

**Note**: the dimension of the array IPIV must be at least $\max(1, \text{N})$.

*On entry*: the pivot indices, as returned by F07BRF (ZGBTRF).

8: ANORM – REAL (KIND=nag_wp) *Input*

*On entry*: if NORM = '1' or 'O', the 1-norm of the **original** matrix $A$.

If NORM = 'I', the $\infty$-norm of the **original** matrix $A$.

ANORM may be computed by calling F06UBF with the same value for the parameter NORM.

ANORM must be computed either **before** calling F07BRF (ZGBTRF) or else from a **copy** of the original matrix $A$ (see Section 9).

*Constraint*: ANORM $\geq 0.0$.

9: RCOND – REAL (KIND=nag_wp) *Output*

*On exit*: an estimate of the reciprocal of the condition number of $A$. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, $A$ is singular to working precision.

10: WORK($2 \times$ N) – COMPLEX (KIND=nag_wp) array *Workspace*

11: RWORK(N) – REAL (KIND=nag_wp) array *Workspace*

12: INFO – INTEGER *Output*

*On exit*: INFO $= 0$ unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO $< 0$

If INFO $= -i$, the $i$th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed estimate RCOND is never less than the true value $\rho$, and in practice is nearly always less than $10\rho$, although examples can be constructed where RCOND is much larger.

## 8 Further Comments

A call to F07BUF (ZGBCON) involves solving a number of systems of linear equations of the form $Ax = b$ or $A^{\mathrm{H}}x = b$; the number is usually 5 and never more than 11. Each solution involves approximately $8n(2k_l + k_u)$ real floating point operations (assuming $n \gg k_l$ and $n \gg k_u$) but takes considerably longer than a call to F07BSF (ZGBTRS) with one right-hand side, because extra care is taken to avoid overflow when $A$ is approximately singular.

The real analogue of this routine is F07BGF (DGBCON).

## 9 Example

This example estimates the condition number in the 1-norm of the matrix $A$, where

$$A = \begin{pmatrix} -1.65 + 2.26i & -2.05 - 0.85i & 0.97 - 2.84i & 0.00 + 0.00i \\ 0.00 + 6.30i & -1.48 - 1.75i & -3.99 + 4.01i & 0.59 - 0.48i \\ 0.00 + 0.00i & -0.77 + 2.83i & -1.06 + 1.94i & 3.33 - 1.04i \\ 0.00 + 0.00i & 0.00 + 0.00i & 4.48 - 1.09i & -0.46 - 1.72i \end{pmatrix}.$$

### 9.1 Program Text

```
    Program f07bufe

!     F07BUF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x02ajf, zgbcon, zgbtrf, zlangb => f06ubf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
      Character (1), Parameter         :: norm = '1'
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: anorm, rcond
      Integer                          :: i, info, j, k, kl, ku, ldab, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:), work(:)
      Real (Kind=nag_wp), Allocatable  :: rwork(:)
      Integer, Allocatable             :: ipiv(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07BUF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kl, ku
      ldab = 2*kl + ku + 1
      Allocate (ab(ldab,n),work(2*n),rwork(n),ipiv(n))
```

```
!     Read A from data file

      k = kl + ku + 1
      Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,n)

!     f06ubf is the NAG name equivalent of the LAPACK auxiliary zlangb
      anorm = zlangb(norm,n,kl,ku,ab(kl+1,1),ldab,rwork)

!     Factorize A
!     The NAG name equivalent of zgbtrf is f07brf
      Call zgbtrf(n,n,kl,ku,ab,ldab,ipiv,info)

      Write (nout,*)
      If (info==0) Then

!       Estimate condition number
!       The NAG name equivalent of zgbcon is f07buf
        Call zgbcon(norm,n,kl,ku,ab,ldab,ipiv,anorm,rcond,work,rwork,info)

        If (rcond>=x02ajf()) Then
          Write (nout,99999) 'Estimate of condition number =', &
            1.0E0_nag_wp/rcond
        Else
          Write (nout,*) 'A is singular to working precision'
        End If
      Else
        Write (nout,*) 'The factor U is singular'
      End If

99999 Format (1X,A,1P,E10.2)
    End Program f07bufe
```

## 9.2 Program Data

```
F07BUF Example Program Data
  4  1  2                                                  :Values of N, KL and KU
 (-1.65, 2.26) (-2.05,-0.85) ( 0.97,-2.84)
 ( 0.00, 6.30) (-1.48,-1.75) (-3.99, 4.01) ( 0.59,-0.48)
               (-0.77, 2.83) (-1.06, 1.94) ( 3.33,-1.04)
                             ( 4.48,-1.09) (-0.46,-1.72)  :End of matrix A
```

## 9.3 Program Results

```
 F07BUF Example Program Results

 Estimate of condition number =  1.04E+02
```